INTRODUCTION TO MuPAD

```
reset:
```

```
2+5;
```

7

```
7!;
```

5040

```
345!
```

242156386507923465587000536919858555701205560402586527348397832670399617201783235\
317473904791361707969553150268947301221382088913488585399281843805644508020148\
636752404948022698231101258810002846873771043764007922001651278559084980475073\
955446603093964326987087311394274684237308398502911304969719715098068025497504\
073058021701657327001169846737892429155078087360515473687954260255463555842826\
903020913423594718635086275165112034783535421871510458382672391689287475258905\
708487655213488727530884968558716385000436989129479527833010340517760688345368\
572902001533686253435387691487120177669920587866285855585726554423099917844925\
480000000000000000000000000000000000000000000000000000000000000000000000000000\
000000

```
(3+I)*(2-I);
```

$7 - i$

```
(x+y)^2;
```

$(x + y)^2$

```
expand((x+y)^2);
```

$x^2 + 2\,x\,y + y^2$

```
factor(x^3-1);
```

$(x - 1)\,(x^2 + x + 1)$

```
factor(x^24-1);
```

$(x - 1)\,(x + 1)\,(x^2 + x + 1)\,(x^2 + 1)\,(x^2 - x + 1)\,(x^4 + 1)\,(x^4 - x^2 + 1)\,(x^8 - x^4 + 1)$

```
sum(k,k=1..100);
```

5050

```
sum(1/k^2,k=1..100);
```

1589508694133037873112297928517553859702383498543709859889432834803818131090369901
/
972186144434381030589657976672623144161975583995746241782720354705517986165248000

```
sum(1/k^2,k=1..1000);
```

```
835459384831496894781878542648548843860444543140864729307638395126038032912078818\
  9588904977469387999844962675327115010933903589145654299730231109091124308462732\
  5329732186766109316261828174601182875501702164588904677785479502529700694366929\
  3307524793996547163688017945296826037413447247331737652629644639707639344639262\
  9796895140901128384286333311745462863716753134735154188954742414035836608258393\
  7099663055379541507590420567361035945849810683329196125645275699319999723182592\
  2036679526675467870525357636249109122511070837028172650873419668453587325849713\
  1645348091123849687614886682117125784781422103460192439394780707024963279033532\
  4685767792564888910543005003079556314194115737948171940383325840598046395049988\
  3029261525552848089894630843538497552630691676216896740675701385847032173192623\
  3881016332493844186817408141003602396236858699094240207812766449/50820720104325\
  1261783529227300076048183979075437485270321545605099258104644816262159803024450\
  0972408259207739139819263052082725188862586270109337163540370629796801206748281\
  2224650586465553482032614190502746121717248161892239954030493982549422690846180\
  5235876956416907687640878308692032203814261826998274713775770604019882671942437\
  3337819478895280853298535971168938897869831095970850418785139173420992068961665\
  5859839289193299599163669641323895022932959750057616390808553697984192067774252\
  3486039845810084061132535320216567518947255952494833022415912350556752737584819\
  8004525569404535304575900241737497049418343827091985156648973444385849478427931\
  1829050180589581507273988682409028088248800576590497216884808783192565859896957\
  2544950280239545397640174350493833629193362885930624768402323396917247538532744\
  70796832851272983644588653710145311847639040000000
```

```
sum(1/k^2,k=1..infinity);
```

$$\frac{\pi^2}{6}$$

```
limit(sin(x)/x,x=0);
```

$$1$$

```
f:=exp(3*cos(5*x));
```

$$e^{3\cos(5x)}$$

```
diff(f,x);
```

$$-15\sin(5x)\,e^{3\cos(5x)}$$

```
diff(f,x,x);
```

$$225\sin(5x)^2\,e^{3\cos(5x)} - 75\cos(5x)\,e^{3\cos(5x)}$$

```
diff(f,x,x,x,x,x);
```

$$843750\sin(5x)^3\,\sigma_1 - 759375\sin(5x)^5\,\sigma_1 - 9375\sin(5x)\,\sigma_1 - 1265625\cos(5x)^2\sin(5x)\,\sigma_1$$

$$+\,2531250\cos(5x)\sin(5x)^3\,\sigma_1 - 421875\cos(5x)\sin(5x)\,\sigma_1$$

where

$$\sigma_1 = e^{3\cos(5x)}$$

```
g:=1/(x^3+1);
```

$$\frac{1}{x^3+1}$$

```
int(g,x);
```

$$\frac{\ln(x+1)}{3} - \frac{\ln\left(\left(x-\frac{1}{2}\right)^2+\frac{3}{4}\right)}{6} + \frac{\sqrt{3}\arctan\left(\frac{2\sqrt{3}\left(x-\frac{1}{2}\right)}{3}\right)}{3}$$

```
int(g,x=0..1);
```

$$\frac{\ln(2)}{3}+\frac{\pi\sqrt{3}}{9}$$

```
plot(g,x=-2..2);
```




```
plot(sin(x),sin(2*x),x=-2*PI..2*PI,LegendVisible);
```




```
plot(sin(a*x),x=-2*PI..2*PI,a=1..2);
```




```
plot(exp(1-x^2-y^2),x=-3..3,y=-3..3,#3D);
```



To define a function ***f*(*x*)** in **MuPAD**, we use arrow notation and enter

```
f:=x->(x^3-7*x^2-x+7)/(50);
```

$$x \rightarrow \frac{x^3 - 7\,x^2 - x + 7}{50}$$

Note that since f is a polynomial, we know that **Domain**(f) is all reals, and because f is of odd
order it follows that **Range**(f) is also all reals. We may plot f over the the interval [-10 , 10] with the following command:

```
plot(f(x),x=-10..10,Colors=[RGB::Blue]);
```



To locate the zeros of f we factor the polynomial expression $f(x)$:

```
factor(f(x));
```

$$\frac{(x-1)\,(x-7)\,(x+1)}{50}$$

Evidently, f has three zeros: $x = -1, 1, 7$. Alternatively, we can find these zeros by writing a **MuPAD** command to solve the equation $f(x) = 0$ for x.

```
s:=solve(f(x)=0,x);
```

$$\{-1,\ 1,\ 7\}$$

In order to refer to the first element in the list of solutions we may enter s [1].

```
s[1];
```

$$-1$$

As you would expect, the third solution is denoted by s[3] in **MuPAD**.

```
s[3];
```

$$7$$

We may sketch a graph of f over a smaller interval to show detailed behavior of the function.

```
plot(f(x),x=-4..8,Colors=[RGB::Black]);
```



Now let us define another function ***g*** in **MuPAD** witht the following command:

```
g:=x->x^4-sqrt(1+x);
```

$x \rightarrow x^4 - \sqrt{1+x}$

Note that **Domain** (***g***) = [-1 , infinity) since is a real number if and only if the argument of the square root is non-negative.

If we try plotting ***g*** over the interval [-1 , 10] we obtain the following graph:

```
plot(g(x),x=-1..10,Colors=[RGB::Blue]);
```



It appears that the zeros of ***g*** occur in the interval (-1 , 5/2) so it makes sense to redraw the graph over a shorter interval.
Doing this will help us to estimate the locations of the zeros.

```
plot(g(x),x=-1..5/2,ViewingBoxYRange=-5..10,
Colors=[RGB::Blue]);
```



Note that the second interval of values in the above **plot** command : **-5 . . 10**, specifies the range of values that appear along the *y* axis.

It is now evident that ***g*** has one zero in the interval (-1 , -1/2) and one in the interval (1 , 3/2).

We will attempt to find these values using a **solve** command as above:

```
s:=solve(g(x)=0,x);
```

$$\left(\{0\} \cup \mathrm{solve}(x^4\,\mathrm{i} < 0, x) \cup \{-\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\} \cup \{-\sigma_1\,\mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\} \cup \{\sigma_1\,\mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\} \cup \{\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z)$$

where

$$\sigma_1 = (y + u\,\mathrm{i})^{1/4}$$

This time the **solve** command does not give us a result that we can use..

In such a case we can utilize the **fsolve** command in the **numerics** package to approximate each of the roots to any desired number of decimal places.

When approximating the zeros of a function ***g*** we use the command

fsolve (g (***x*** **) = 0 ,** ***x*****= a . . b) ;**

where (a , b) is an interval known to contain a single root to the equation $g(x) = 0$.

```
numeric::fsolve(g(x),x=-1..-1/2);
```

$$[x = -0.81165232]$$

We can also return just the numerical value. Here we are getting the **fsolve** routine, computing and storing the numerical value, and then displaying the value.

```
use(numeric, fsolve);
```

```
xvalue:=fsolve(g(x),x=-1..-1/2)[1][2]:
```

```
xvalue;
```

$$-0.81165232$$

```
fsolve(g(x)=0,x=1..3/2);
```

$$[x = 1.096981558]$$

Finally, to determine the range of ***g***, we observe from our graph that **Ran**(*g*) contains all numbers larger than the minimum value of g(x).In class we will develop an analytic method for finding this minimum . For now, we can estimate the minimum value of $g(x)$ by redrawing the graph of g over the short interval

[0.50 , 0.80].

```
plot(g(x),x=-.5..0.80,Colors=[RGB::Blue]);
```



Now position the mouse arrow inside the coordinate plane of this plot and click the left hand mouse

button once. Now move the mouse arrow to the minimum point and click once again and hold . A tiny window will appear showing the x- and y- coordinate values at the location of the cursor. The y-coordinate of this point approximates the minimum value of g.

The actual minimum value is -1.163640263 , where .

```
xloc:=fsolve(D(g)(x)=0,x=0..1);
```

$[x = 0.4689591882]$

```
xloc[1][2];
```

0.4689591882

```
gmin:=g(xloc[1][2]);
```

-1.163640263

Thus, **Ran** (***g***) = [gmin, infinity).

Composition of Functions

We can define the composition of the function f with g in MuPAD with the following command:

```
h:=x->(f@g)(x);
```

$x \to (f \circ g)(x)$

To see the value of h(x) we enter

```
h(x);
```

$$\frac{\sqrt{x+1}}{50} - \frac{\left(\sqrt{x+1} - x^4\right)^3}{50} - \frac{7\left(\sqrt{x+1} - x^4\right)^2}{50} - \frac{x^4}{50} + \frac{7}{50}$$

If we would like to expand the above expression into a sum of simpler terms
we may enter the following command where the parentheses (%) represents
the results of the immediately preceding command (stands for "the last output").

Following is first the graph of h, and the graphs of f , g , and h plotted on the same coordinate system.

```
plot(h(x),x=-1..2,ViewingBoxYRange=-2..5,
Colors=[RGB::Blue]);
```




```
plot(f(x),g(x),h(x),x=-1..2,ViewingBoxYRange=-2..5,
Colors=[RGB::Red,RGB::Green,RGB::Blue]);
```



Solving Inequalities With Graphics

The graphic capability of **MuPAD** can also be used to solve inequalities.
As an example, consider the inequality $x^2 < \cos(x)\,\sin(x)$ for x in [- π , π].
First graph the two functions together.

```
f:=x->x^2; g:=x->sin(x)*cos(x);
```

```
plot(f(x),g(x),x=-PI..PI,Colors=[RGB::Red,RGB::Blue]);
```

$x \to x^2$

$x \to \sin(x)\cos(x)$



The solution to the inequality is the set of numbers x such that f(x) < g(x), i. e. , those values of x such that the green curve is above the red curve.
We will look more closely at the interval [0,1]:

```
plot(f(x),g(x),x=0..1,Colors=[RGB::Red,RGB::Blue]);
```



The curves intersect at 0 and at a point with X-coordinate between 0.6 and 0.8.
To find this point we will solve the equation f(x)=g(x):

```
solve(f(x)=g(x),x);
```

$\text{solve}(x^2 - \cos(x)\sin(x) = 0,\ x)$

Again we need to use "**fsolve**".

```
fsolve(f(x)=g(x),x=0.6..0.8);%[1][2];
```

$[x = 0.702207412]$

0.702207412

Therefore the solution set is the open interval $(0, 0.702207412)$.

The operands of an object - op

`op(object)` returns all operands of the object.

`op(object, i)` returns the `i`-th operand.

```
a:=x^24-1;
```
$$x^{24}-1$$

```
b:=factor(a)
```
$$(x-1)\,(x+1)\,(x^2+x+1)\,(x^2+1)\,(x^2-x+1)\,(x^4+1)\,(x^4-x^2+1)\,(x^8-x^4+1)$$

```
c:=(x-2)^2*(x-4)*(x+2)
```
$$(x-2)^2\,(x+2)\,(x-4)$$

```
l:=m/n;
```
$$\frac{m}{n}$$

```
op(a)
```
$$x^{24},\ -1$$

```
op(b)
```
$$1,\ x-1,\ 1,\ x+1,\ 1,\ x^2+x+1,\ 1,\ x^2+1,\ 1,\ x^2-x+1,\ 1,\ x^4+1,\ 1,\ x^4-x^2+1,\ 1,\ x^8-x^4+1,\ 1$$

```
op(b,4)
```
$$x+1$$

```
op(c)
```
$$(x-2)^2,\ x+2,\ x-4$$

```
c[2]
```
$$x+2$$

```
op(l)
```
$$m,\ \frac{1}{n}$$

```
l[2]
```
$$n$$

```
op(l,2)
```
$$\frac{1}{n}$$

```
nops(c)
```
$$3$$

<u>OP</u>

```
op(2*c^t+9/v);
```
$$2\,((x-2)^2\,(x+2)\,(x-4))^t,\ \frac{9}{v}$$

```
op(5*h+9/j-4^x*y+6-2*R);
```
$$-2\,R,\ -4^x\,y,\ \frac{9}{j},\ 5\,(x\rightarrow(f\circ g)(x)),\ 6$$

```
type(5*h+9/j-4^x*y+6-2*R);
```

"_plus"

```
nops(5*h+9/j-4^x*y+6-2*R);
```

5

```
op(sin(x),0);
```

sin

```
op(5*h+9/j-4^x*y+6-2*R,3);
```

$\frac{9}{j}$

```
op(t/s);
```

$t, 1/\left(\left(\{0\} \cup \mathrm{solve}(x^4\, \mathrm{i} < 0, x) \cup \left\{-\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{-\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z)\right)$

where

$\sigma_1 = (y + u\, \mathrm{i})^{1/4}$

```
op(t/s,2);
```

$1/\left(\left(\{0\} \cup \mathrm{solve}(x^4\, \mathrm{i} < 0, x) \cup \left\{-\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{-\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z)\right)$

where

$\sigma_1 = (y + u\, \mathrm{i})^{1/4}$

```
op(1/s);
```

$\left(\{0\} \cup \mathrm{solve}(x^4\, \mathrm{i} < 0, x) \cup \left\{-\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{-\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z), -1$

where

$\sigma_1 = (y + u\, \mathrm{i})^{1/4}$

```
op(1/s,1);
```

$\left(\{0\} \cup \mathrm{solve}(x^4\, \mathrm{i} < 0, x) \cup \left\{-\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{-\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1\, \mathrm{i} \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1 \mid u \in \mathbb{R} \wedge y \in (0, \infty)\right\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z)$

where

$\sigma_1 = (y + u\, \mathrm{i})^{1/4}$

```
op(1/s,2);
```

-1

```
a:=7*x*y-9*z+8/p*o^3;
```

$7\, x\, y - 9\, z + \frac{8\, o^3}{p}$

```
op(op(op(a,3),2),1);
```

p

```
b1:=op(a,3);
```

$$\frac{8\,o^3}{p}$$

```
b2:=op(b1,2);
```

$$\frac{1}{p}$$

```
b3:=op(b2,1);
```

p

SUBSOP

```
a:=3*t*s+5*p/u-9;
```

$$\left\{3\,t\,\mathrm{z1}+\frac{5\,p}{u}-9 \;\middle|\; \mathrm{z1} \in \left(\{0\} \cup \mathrm{solve}(x^4\,\mathrm{i} < 0, x) \cup \left\{-\sigma_1 \;\middle|\; x \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{-\sigma_1\,\mathrm{i} \;\middle|\; x \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1\,\mathrm{i} \;\middle|\; x \in \mathbb{R} \wedge y \in (0, \infty)\right\} \cup \left\{\sigma_1 \;\middle|\; x \in \mathbb{R} \wedge y \in (0, \infty)\right\}\right) \cap \mathrm{RootOf}(z^8 - z - 1, z)\right\}$$

where

$$\sigma_1 = (y + x\,\mathrm{i})^{1/4}$$

```
op(op(op(a,2),2),1);
```

FAIL

```
subsop(a,[2,2,1]=(Grisha))
```

FAIL

Question 1:

Given y=(2x+3)e^{3x} , find the value of y when dy/dx=2.
Give your answer to five significant figures.

Answer 1 Clumsy code, no comments

```
reset():
```

```
solve(diff((2*x+3)*exp(3*x),x)=2,x);
```

$$\left\{\frac{\mathrm{W}_k\left(\mathrm{e}^{11/2}\right)}{3} - \frac{11}{6} \;\middle|\; k \in \mathbb{Z}\right\}$$

```
numeric::solve(diff((2*x+3)*exp(3*x),x)=2,x);
```

$\{-0.4696102659\}$

```
(2*x+3)*exp(3*x)|x=%[1];
```

0.5037140647

Answer 2

```
reset():
```

First assign y and find its derivative

```
y:=(2*x+3)*exp(3*x);
```

$$\mathrm{e}^{3\,x}\,(2\,x+3)$$

```
deriv:=diff(y,x);
```

$2\,e^{3\,x} + 3\,e^{3\,x}\,(2\,x + 3)$

Now solve to find x when dy/dx =2

```
xval:=numeric::solve(deriv=2,x);
```

$\{-0.4696102659\}$

Evaluate y

```
yval:=y|x=xval;
```

$\{0.5037140647\}$

and get the answer in a suitable form

```
DIGITS:=5: float(yval);
```

$\{0.50371\}$

So when dy/dx=2, y=0.50374 to five significant figures